CASE STUDY

twenty one

Beta of Stocks

case study

OVERVIEW

CS21.1 *Application Overview and Model Development*

The "Beta of Stocks" application seeks to predict the return on several stocks based on the return of the market. In this application, the term beta refers to the measure of a stock or portfolio's volatility in comparison to the market as a whole. We provide the users with two options for estimating the beta values necessary in making these predictions:

Option 1: Determine the optimal beta value per stock for a given period of time.
Option 2: Evaluate the MSE (mean square error) for various values over the past months for the beta estimate.

CS21.1.1 Model Definition and Assumptions

The users begin the application by selecting a portfolio of stocks from a set database. We provide historical data about the stock returns for each stock in the database. We use S&P data for the corresponding market return values in this historical data. (S&P is the Standard and Poors; a financial services company that rates stocks according to risk profiles and maintains an index consisting of 500 stocks which reflects the risk/return characteristics of the market). The users, then, are seeking to determine the beta values based on the two optional estimation methods so they can better predict future stock returns for their portfolios.
The underlying model for predicting the return on a stock is:

$$r_s = \alpha + \beta * r_m$$

Where r_s is the return of the stock, we are predicting, r_m is the return on the market, and α and β are parameters. The β parameter is what we refer to as the beta of a stock, which measures the responsiveness of a stock's returns to the market return.

For Option 1, to estimate each stock's α and β parameters, we use the Solver to calculate those that minimize the MSE between the actual stock returns and the predicted stock returns. The users provide an input that directs which months of historical data to include while determining the best beta estimate. This value is indicated as an interval using the month numbers in the historical data.

For Option 2, we also consider the number of months of historical data used in determining the optimal beta value to be an important factor. Therefore, we must determine the best number of months to minimize the MSE of the stock return predictions. We estimate the beta parameter with the SLOPE function and calculate the average beta for the resulting best number of months. The users provide an input of the number of trial months with which to experiment. This input is an interval of the number of months to use with an assumed step size of one month.

These calculations will become clear as we describe the worksheets and procedures. For more information, please see *Practical Management Science* by Winston and Albright.

CS21.1.2 Input

The general input for this application is:

- the portfolio of stocks.

The input for Option 1 is:

- which months of historical data to use.

The input for Option 2 is:

- the range of trial values for the number of months to use.

CS21.1.3 Output

The output for Option 1 is:

- the optimal beta value for each stock;
- each stock's minimized MSE value for this optimal beta; and
- the chart of the predicted stock returns versus the historical stock returns for each stock.

The output for Option 2 is:

- the best number of past months to use;
- the minimized MSE for the best number of months used;
- the mean beta value for the best number of months used;
- the sum of MSE values for the entire portfolio;
- the chart of MSE values for all stocks; and
- a histogram of beta values for the best number of months used for each stock.

CS21.2 *Worksheets*

We incorporate five worksheets in this application: the welcome sheet, two calculation sheets, and two option sheets. The welcome sheet includes the title, the description of the application, and some images (see Figure CS21.1). The "Start" button on the welcome sheet brings the users to a form where they make their portfolio selection. The users then see an option form where they can decide to use Option 1 or Option 2 for determining the beta estimates.

Figure CS21.1 The welcome sheet.

If the users choose Option 1, then the output sheet for Option 1 appears (see Figure CS21.2). Here, using the controls on the worksheet, they can specify which months of historical data they want to use to determine the optimal beta values. They then click the "Calculate" button to run the calculations and view the results. They can view the "Stock Performance" chart for each stock in their portfolio using the combo box above the chart.

The “Return to Menu” button returns the users to the options form. The “End” button allows them to exit the application.



Figure CS21.2 The option 1 output sheet.

The calculations for Option 1 are performed on the “Calc 1” sheet (see Figure CS21.3), which always remains hidden from the users. The first columns on this sheet are the S&P market returns (see Figure CS21.3(a)) over the past 146 months. The next columns are for all of the stocks in the database. We have two columns for each stock: the actual returns from the historical data; and the predicted returns. The latter are determined using the formula:

$=Alpha_s + MarketReturns * Beta_s$

We are solving for each stock’s alpha and beta values. To do so, we keep a set of ranges named “$Alpha_s$” and “$Beta_s$” for each stock and reference them in these columns (see Figure CS21.3(b)). For example, since the “Cin” stock is listed first in the database of stocks, the formula for predicting its returns is:

=Alpha1+ MarketReturns * Beta1

We also include a column for calculating the square error between the historical returns and the predicted returns. We update this column’s formulas depending on which stock we are currently solving for. The MSE value is the average of these values for the rows corresponding to the historical months specified by the users to use. We also include some columns for copying the historical and predicted returns for the source data of the “Stock Performance” chart. We copy the data for the stock selected from the combo box and paste it above the chart on the Option 1 output sheet.



Option 1 Calculations

END

Actual and Predicted Stock Returns

Month	Date	S and P	Cin	Cin**	Dell	Dell**	INTC	IN
1	Sep-88	0.04	0.009470175	0.027843404	0.282112845	0.131398923	-0.017828571	0.07
2	Oct-88	0.03	0.034972764	0.020387494	0.1582397	0.102633192	-0.100069816	0.05
3	Nov-88	-0.01	-0.037032799	8.0057E-06	-0.084074373	0.024006862	-0.19188001	0.00
4	Dec-88	0.02	0.009610942	0.015913948	-0.080317741	0.085373754	0.18752	0.04
5	Jan-89	0.07	0.012383132	0.042755224	-0.049904031	0.188930385	0.094718405	0.12
6	Feb-89	-0.03	-0.004804836	-0.005459662	-0.171717172	0.002911992	-0.0096	-0.01
7	Mar-89	0.02	-0.028981181	0.018896312	-0.095121951	0.096880046	0.019386107	0.05
8	Apr-89	0.05	0.047916917	0.032814011	0.105121294	0.150576077	0.104717786	0.09
9	May-89	0.04	0.029119155	0.026849283	0.079268293	0.127563492	0.120723902	0.07
10	Jun-89	-0.01	0.042461424	0.004481552	-0.073446328	0.0412663	-0.107719575	0.01
11	Jul-89	0.09	0.053001391	0.051702317	-0.142682927	0.223449262	0.05175458	0.14
12	Aug-89	0.02	-0.008761531	0.016411008	0.036984353	0.087291469	-0.016367642	0.04
13	Sep-89	0.00	0.035390065	0.004978613	0.017832647	0.043184015	0.074986667	0.01
14	Oct-89	-0.02	0.037579758	-0.00446554	-0.157681941	0.006747423	0.046537011	-0.01
15	Nov-89	0.02	0.029411765	0.01740513	-0.0416	0.0911269	0.029581872	0.05
16	Dec-89	0.02	0.028571429	0.018896312	-0.043405676	0.096880046	-0.007182982	0.05
17	Jan-90	-0.07	-0.008581671	-0.02633621	-0.158813264	-0.077632054	0.144977275	-0.07

Figure CS21.3 (a)



MSE
0.00791595

FE	PFE**	Square Error
411073	0.07912763	
588773	0.056002237	
502857	-0.007207169	
035278	0.042127002	
345547	0.125378415	
400021	-0.02416579	
100586	0.051377159	
062165	0.094544558	
619362	0.076044244	
943588	0.006668066	
512175	0.153128886	
400747	0.043668695	0.0040933
041357	0.008209759	0.00350728
537041	-0.021082405	0.00325823
722247	0.04675208	0.00043259
394739	0.051377159	0.00380905
412796	-0.08891689	0.04684874
117813	0.034418538	0.00041399
569673	0.054460545	0.00040066
915622	-0.02416579	0.00164583

Stock	Alpha	Beta
Cin	0.01	0.50
Dell	0.05	1.92
INTC	0.02	1.38
MSFT	0.03	0.99
NT	0.00	1.06
PFE	0.01	1.54

Chart Data

12	-0.008761531	0.016411008
13	0.035390065	0.004978613
14	0.037579758	-0.00446554
15	0.029411765	0.01740513
16	0.028571429	0.018896312
17	-0.008581671	-0.02633621
18	0.00408052	0.013428644
19	-0.028457615	0.019890433
20	-0.030492224	-0.005459662
21	0.048463748	0.055678802
22	0.012599949	0.00348743
23	0.058161993	0.005475673
24	-0.072000304	-0.037768606
25	-0.008615038	-0.017389118
26	0.051023466	0.004978613
27	0.016870336	0.038778739
28	-0.008295225	0.020387494
29	0.008647331	0.028837525
30	0.046606484	0.042755224
31	0.044540283	0.018896312
32	0.003421539	0.008458037

Figure CS21.3 (b)

Figure CS21.3 The calculation sheet for Option 1.

If the users select Option 2, they will then see the Option 2 output sheet (see Figure CS21.4). The users enter an upper and lower bound for the trial number of months to use in estimating beta and then click the “Calculate” button. The calculations are performed and the optimal number of historical months is recorded for each stock. The minimum MSE and the mean beta value for this number of months are also recorded. The sum of the MSE for the entire portfolio displays above a bar graph.

The users can view two chart types: the MSE values for all the stocks in the portfolio or a histogram of the beta values for each stock. To select which chart to view, the users can click on the option buttons above the chart. If the users are viewing the histogram of beta values for a stock, they can select which stock to view from the combo box above the chart. As in the previous output sheet, the “Return to Menu” button returns the users to the options form, and the “End” button allows them to exit the application.



Figure CS21.4 The option 2 output sheet.

The calculations for Option 2 are performed on the "Calc 2" sheet (see Figure CS21.5). The first two columns again are the historical S&P market returns for 146 months (see Figure CS21.5 (a)). The actual stock returns for all of the stocks in the database are also recorded. We then include a column that calculates the beta values of a particular stock with the SLOPE formula:

=SLOPE(StockReturns / MarketReturns)

The number of months of stock returns and market returns selected for this formula will vary as we try all of the values in the interval provided by the users. For each beta value, we calculate the predicted stock return with the formula:

=MarketReturns * $Beta_s$

This is the same formula as in Option 1, with $Alpha_s$ assumed to be 0. Using this predicted value and the actual stock returns, we calculate the square errors. From these errors, we determine the MSE.

For each trial value of the number of historical months to use in the calculations, we record the MSE and the mean beta value (see Figure CS21.5 (b)). We keep these values separate for the chart source data of the two charts available on the output sheet. We calculate the histogram data for each selected stock by referring to the mean beta column.



Option 2 Calculations | END

MSE 0.015

Actual Stock Returns

Date	S and P	Cin	Dell	INTC	MSFT	NT	PFE	Beta	Predicted	Square Error
Sep-88	0.04	0.00947	0.282113	-0.01783	0.045075	0.058011	0.071411			
Oct-88	0.03	0.034973	0.15824	-0.10007	-0.06215	-0.02744	0.050589			
Nov-88	-0.01	-0.03703	-0.08407	-0.19188	-0.03585	-0.06334	0.002503			
Dec-88	0.02	0.009611	-0.08032	0.18752	0.127095	0	0.022035			
Jan-89	0.07	0.012383	-0.0499	0.094718	0.119659	-0.06019	0.003346			
Feb-89	-0.03	-0.0048	-0.17172	-0.0096	-0.00205	-0.00796	-0.0434			
Mar-89	0.02	-0.02898	-0.09512	0.019386	-0.16179	-0.03225	0.043101			
Apr-89	0.05	0.047917	0.105121	0.104718	0.120254	0.16664	0.076062			
May-89	0.04	0.029119	0.079268	0.120724	0.082861	-0.00715	-0.00562			
Jun-89	-0.01	0.042461	-0.07345	-0.10772	-0.124	0.021615	-0.05944			
Jul-89	0.09	0.053001	-0.14268	0.051755	0.033012	0.13377	0.143512			
Aug-89	0.02	-0.00876	0.036984	-0.01637	0.073119	0.105626	0.065401			
Sep-89	0.00	0.03539	0.017833	0.074987	0.165931	-0.03935	-0.00904			

Figure CS21.5 (a)



MSE Values per Stock

Months	Cin	INTC					Months	Mean Beta
20	0.003166	0.012005					20	1.33936
21	0.00317	0.011718					21	1.35015
22	0.003216	0.011619					22	1.35598
23	0.003167	0.011618					23	1.36116
24	0.003216	0.011319					24	1.36429
25	0.003246	0.011278					25	1.36457
26	0.003233	0.011287					26	1.364
27	0.003266	0.011383					27	1.3649
28	0.003291	0.011402					28	1.36493
29	0.003319	0.011274					29	1.36688
30	0.003343	0.011432					30	1.36853
31	0.003369	0.011554					31	1.36965
32	0.003399	0.011643					32	1.37168
33	0.003428	0.011703					33	1.37224
34	0.003443	0.011678					34	1.3726
35	0.003482	0.01178					35	1.37363

Beta Histogram

Bin	Frequency	umulative %
-.473	1	.79%
-.338	5	4.76%
-.202	8	11.11%
-.067	10	19.05%
.068	2	20.63%
.203	0	20.63%
.339	16	33.33%
.474	40	65.08%
.609	15	76.98%
.745	7	82.54%
.880	14	93.65%
More	8	100.00%

Figure CS21.5 (b)
Figure CS21.5 The calculation sheet.

Summary

Welcome Sheet	The application description and the "Start" button.
Option 1 Output Sheet	Input areas for which months to use; the optimal beta and minimized MSE for each stock; the chart of stock performance for each stock.
Option 1 Calculation Sheet	Historical data for market and stocks; Solver determines $Alpha_s$ and $Beta_s$ for each stock; chart data.
Option 2 Output Sheet	Input areas for bounds on the trial number of months to use; optimal number of months and corresponding MSE and mean beta; sum of MSE for portfolio; MSE chart for all the stocks; histogram of beta values for a selected stock.
Option 2 Calculation Sheet	Historical data; beta calculations using SLOPE function; record of MSE and mean beta for each trial; chart data and histogram data.

CS21.3 *User Interface*

This application includes two user forms, controls on both of the output sheets, and several navigational and functional buttons. The first form that the users observe is the

portfolio form (see Figure CS21.6). In this form, we use two list boxes to reveal the stocks available in the database and the stocks in the users' portfolio. We include two command buttons (with images instead of captions) to enable the users to add stocks to or remove stocks from their portfolio. We add extra labels to ensure that control functionality is clear.



Figure CS21.6 The portfolio form.

The second form that the users see is the options form (see Figure CS21.7). We include two option buttons in a frame to enable the users to select between Option 1 and Option 2. Depending on their choice, they will be taken to the corresponding output sheet.



Figure CS21.7 The options form.

The controls on the Option 1 output sheet include two text boxes, an option button, and a combo box (see Figure CS21.2). The text boxes enable the users to specify an interval of months to use for the Option 1 calculations. The option button indicates the active chart and the combo box enables the users to select which stock is graphed in the chart.

The controls on the Option 2 output sheet include two text boxes, two option buttons, and a combo box (see Figure CS21.4). The text boxes enable the users to specify upper and lower bounds for the trial number of months for the Option 2 calculations. The option

button indicates which chart is active, and the combo box enables the users to select which stock is graphed in the second chart.

Navigational buttons include “End” and “Return to Menu.” Functional buttons include “Calculate” and “Start.”

Summary

Portfolio Form	Users create their portfolio by clicking on buttons to add stocks to and remove stocks from the stocks and portfolio lists.
Options Form	Users choose between Option 1 and Option 2 with option buttons.
Controls on Option 1 Output Sheet	Include: two textboxes for input values, one option button, and a combo box for charts.
Controls on Option 2 Output Sheet	Include: two textboxes for input values, two option buttons, and a combo box for charts.
Navigational Buttons	“End,” “Return to Menu.”
Functional Buttons	“Calculate,” “Start.”

CS21.4 *Procedures*

We will now outline the procedures for this application, beginning with the initial sub procedures and variable definitions (see Figure 28.13). The *Main* procedure initializes certain variables and calls the *ClearPrev* procedure, which clears previous values from all sheets. The *Main* procedure then displays the portfolio form to the users. After the users create their portfolio, the *PrepCalc* procedure is called. This procedure prepares the output and calculation sheets based on the stocks that the users had selected (see Figure 28.14). The *Main* procedure then presents the options form.

The code for the portfolio form sets a Boolean value in a *Portfolio* array to True when the stock is added and to false when it is removed (see Figure CS21.10). These procedures also perform some error checking to ensure that stocks are not added multiple times and that the users are operating the controls correctly. The options form code takes the users to the appropriate output sheet depending on their selection (see Figure CS21.11).

If the users press the “Calculate” button on the Option 1 output sheet, the *DoOption1* procedure is called (see Figure CS21.12). This procedure begins by recording the input specified by the users on the output sheet’s controls. It then loops over the *Portfolio* array to update the formulas and run the Solver for each stock selected by the users. When the calculations are done, the values are recorded on the output sheet. The code then calls the Worksheets(“Option1”).cmbStocks_Change event procedure to update the chart data.

The procedures for the “Option1” output sheet are event procedures for the option button and combo box (see Figure CS21.13). These procedures update the chart data on the Option 1 calculation sheet.

If the users select the “Calculate” button on the Option 2 output sheet, the *DoOption2* procedure is called (see Figure CS21.14). This procedure begins by recording the input specified by the users on the output sheet’s controls. It then loops from the lower to the

upper bound of the number of months to use in the SLOPE formula and updates the formula for each loop.

```
Option Explicit
Option Base 1

Public i As Integer, Portfolio(6, 2) As Variant, NumStocks As Integer, _
MonStart As Integer, MonEnd As Integer, NumMon As Integer, s As Integer, CurrStock As Integer, _
CalcDone As Boolean

Sub Main()
    For i = 1 To 6
        Portfolio(i, 1) = Range("AllStocks").Cells(i)
        Portfolio(i, 2) = False
    Next i
    CalcDone = False
    Call ClearPrev

    frmPortfolio.Show
    Call PrepCalc
    frmOptions.Show
End Sub

Sub ClearPrev()
    With Worksheets("Calc1")
        .Range(.Range("Error1").Offset(1, 0), .Range("Error1").End(xlDown)).ClearContents
        .Range(.Range("Opt1Chart"), .Range("Opt1Chart").Offset(0, 2).End(xlDown)).ClearContents
    End With

    With Worksheets("Calc2")
        .Range(.Range("ChartStocks"), .Range("ChartStocks").Offset(0, 5)).ClearContents
        .Range(.Range("BetaEst").Offset(1, 0), .Range("Error2").End(xlDown)).ClearContents
        .Range(.Range("ChartMonths").Offset(1, 0), .Range("ChartMonths").Offset(1, 8).End(xlDown)).ClearContents
        .Range(.Range("HistOutput"), .Range("HistOutput").Offset(20, 2)).Clear
    End With

    With Worksheets("Option1")
        .Range(.Range("StockName").Offset(1, 0), .Range("StockName").Offset(6, 2)).ClearContents
    End With

    With Worksheets("Option2")
        .Range(.Range("StockName2").Offset(1, 0), .Range("StockName2").Offset(6, 3)).ClearContents
    End With
End Sub
```

Figure CS21.8 Variable declarations, Main procedure, and *ClearPrev* procedure.

```
Sub PrepCalc()
    NumStocks = 0
    For i = 1 To 6
        If Portfolio(i, 2) = True Then
            NumStocks = NumStocks + 1
            Range("ChartStocks").Offset(0, NumStocks - 1).Value = Portfolio(i, 1)
            Range("StockName").Offset(NumStocks, 0).Value = Portfolio(i, 1)
            Range("StockName2").Offset(NumStocks, 0).Value = Portfolio(i, 1)
        End If
    Next i

    Range(Range("StockName2").Offset(1, 0), Range("StockName2").Offset(NumStocks, 0)).Name = "StockList2"
End Sub
```

Figure CS21.9 The *PrepCalc* procedure.

```
Private Sub cmdCancel_Click()
    Unload Me
    End
End Sub

Private Sub cmdOK_Click()
    Unload Me
End Sub

Private Sub cmdAdd_Click()
    For i = 1 To 6
        If Portfolio(i, 1) = lstAll.Value Then
            If Portfolio(i, 2) = True Then
                MsgBox "You have already added this stock."
                Exit Sub
            Else
                Portfolio(i, 2) = True
            End If
        End If
    Next i
    lstPort.AddItem lstAll.Value
End Sub

Private Sub cmdRemove_Click()
    For i = 1 To 6
        If Portfolio(i, 1) = lstPort.Value Then
            Portfolio(i, 2) = False
        End If
    Next i
    On Error GoTo DisplayError
    lstPort.RemoveItem lstPort.ListIndex
    Exit Sub
DisplayError:
    MsgBox "This action will cause an error. Please re-read the instructions and try again."
End Sub

Private Sub UserForm_Initialize()
    lstAll.RowSource = "AllStocks"
    lstAll.Value = lstAll.List(0)
End Sub
```

Figure CS21.10 The portfolio form code.

```
Private Sub cmdCancel_Click()
    Unload Me
    End
End Sub

Private Sub cmdOK_Click()
    If opt1 Then
        Worksheets("Option1").Visible = True
    ElseIf opt2 Then
        Worksheets("Option2").Visible = True
    End If

    Worksheets("Welcome").Visible = False
    Unload Me
End Sub
```

Figure CS21.11 The options form code.

```
Sub DoOption1()
    Application.ScreenUpdating = False
    With Worksheets("Option1")
        MonStart = CInt(.txtMonStart)
        MonEnd = CInt(.txtMonEnd)
    End With
    Range(Range("Error1").Offset(1, 0), Range("Error1").End(xlDown)).ClearContents
    Range(Range("Opt1Chart"), Range("Opt1Chart").Offset(0, 2).End(xlDown)).ClearContents

    Worksheets("Calc1").Activate
    NumStocks = 0
    For i = 1 To 6
        If Portfolio(i, 2) = True Then
            NumStocks = NumStocks + 1
            Range(Range("Error1").Offset(MonStart, 0), Range("Error1").Offset(MonEnd, 0)).FormulaR1C1 = _
                "=(RC[-" & 14 - (2 * i) & "] - RC[-" & 15 - (2 * i) & "])^2"
            Range("ObjFunc").Formula = "=AVERAGE(" & Range("Error1").Offset(MonStart, 0).Address & ":" & _
                Range("Error1").Offset(MonEnd, 0).Address & ")"

            SolverReset
            SolverOK SetCell:=Range("ObjFunc"), MaxMinVal:=2, ByChange:=Range(Range("Alpha" & i), Range("Beta" & i))
            SolverAdd CellRef:=Range(Range("Alpha" & i), Range("Beta" & i)), Relation:=1, FormulaText:=1
            SolverOptions AssumeNonNeg:=True
            SolverSolve UserFinish:=True

            Range("OptBeta").Offset(NumStocks, 0).Value = Range("Beta" & i).Value
            Range("OptMSE").Offset(NumStocks, 0).Value = Range("ObjFunc").Value
        End If
    Next i
    CalcDone = True
    With Worksheets("Option1")
        .optStock = True
        .cmbStocks.Value = .cmbStocks.List(0)
        Call .cmbStocks_Change
    End With
    Application.ScreenUpdating = True
End Sub
```

Figure CS21.12 *DoOption1* procedure.

```
Sub cmbStocks_Change()
    If CalcDone Then
        Application.ScreenUpdating = False
        Dim SelectedStock As Integer
        SelectedStock = Application.WorksheetFunction.Match(cmbStocks.Value, Worksheets("Calc1").Range("Opt1Stocks"), 0)

        With Worksheets("Calc1")
            .Range(.Range("MonthStart").Offset(MonStart, 0), .Range("MonthStart").Offset(MonEnd, 0)).Copy
            .Range("Opt1Chart").PasteSpecial
            .Range(.Range("StockStart").Offset(MonStart, SelectedStock), .Range("StockStart").Offset(MonEnd, SelectedStock + 1)).Copy
            .Range("Opt1Chart").Offset(0, 1).PasteSpecial xlPasteValues

            Worksheets("Option1").Activate
            ActiveSheet.ChartObjects("Chart1").Select
            ActiveChart.SetSourceData Source:=.Range(.Range("Opt1Chart"), .Range("Opt1Chart").Offset(0, 2).End(xlDown))
            ActiveChart.SeriesCollection(1).Name = "Actual"
            ActiveChart.SeriesCollection(2).Name = "Predicted"
            ActiveChart.ChartTitle.Characters.Text = cmbStocks.Value
        End With
        Range("A1").Select
        Application.ScreenUpdating = True
    End If
End Sub

Sub optStock_Click()
    optStock = True
    cmbStocks.Value = cmbStocks.List(0)
    Call cmbStocks_Change
End Sub
```

Figure CS21.13 The Option 1 output sheet code.

Once the loop is complete, it searches for the number of months that yields the minimum MSE value for each stock (see Figure CS21.15). It then updates the output sheet and the MSE chart accordingly.

```
Sub DoOption2()
    Application.ScreenUpdating = False
    With Worksheets("Option2")
        MonStart = CInt(.txtMonStart)
        MonEnd = CInt(.txtMonEnd)
    End With
    Range(Range("BetaEst").Offset(1, 0), Range("Error2").End(xlDown)).ClearContents
    Range(Range("ChartMonths").Offset(1, 0), Range("ChartMonths").Offset(1, 8).End(xlDown)).ClearContents
    Range(Range("HistOutput"), Range("HistOutput").Offset(20, 2)).Clear

    NumMon = MonStart
    Do While NumMon <= MonEnd
        NumStocks = 0
        Range(Range("Predict").Offset(NumMon + 1, 0), Range("Predict").Offset(146, 0)).FormulaR1C1 = "=RC[-1]*RC[-10]"
        Range("MSE").Formula = "=AVERAGE(" & Range("Error2").Offset(NumMon + 1, 0).Address & ":" & _
            Range("Error2").Offset(146, 0).Address & ")"
        For i = 1 To 6
            If Portfolio(i, 2) = True Then
                NumStocks = NumStocks + 1
                Range(Range("BetaEst").Offset(NumMon + 1, 0), Range("BetaEst").Offset(146, 0)).FormulaR1C1 = _
                    "=SLOPE(R[-1]C[" & i - 8 & "]:R[-" & NumMon & "]C[" & i - 8 & "], R[-1]C[-9]:R[-" _
                    & NumMon & "]C[-9])"
                Range(Range("Error2").Offset(NumMon + 1, 0), Range("Error2").Offset(146, 0)).FormulaR1C1 = _
                    "=(RC[-1] - RC[" & i - 10 & "])^2"

                Range("ChartMonths").Offset(NumMon - MonStart + 1, 0).Value = NumMon
                Range("ChartStocks").Offset(NumMon - MonStart + 1, NumStocks - 1).Value = Range("MSE").Value
                Range("MeanBeta").Offset(NumMon - MonStart + 1, 0).Value = _
                    Application.WorksheetFunction.Average(Range(Range("BetaEst").Offset(NumMon + 1, 0), _
                    Range("BetaEst").Offset(146, 0)))
            End If
        Next i
        NumMon = NumMon + 1
        Range(Range("BetaEst").Offset(1, 0), Range("Error2").End(xlDown)).ClearContents
    Loop
```

Figure CS21.14 The first part of the *DoOption2* procedure.

```
    Range(Range("ChartMonths").Offset(1, 0), Range("ChartMonths").End(xlDown)).Copy
    Range("ChartMonths").Offset(1, 7).PasteSpecial
    For i = 1 To NumStocks
        Range("MinMSE2").Offset(i, 0).Value = Application.WorksheetFunction.Min(Range(Range("ChartStocks").Offset(1, i - 1), _
            Range("ChartStocks").Offset(1, i - 1).End(xlDown)))
        Range("MinMSE2").Offset(i, -1).Value = Application.WorksheetFunction.VLookup(Range("MinMSE2").Offset(i, 0).Value, _
            Range(Range("ChartStocks").Offset(1, i - 1), Range("ChartStocks").Offset(1, 6).End(xlDown)), 8 - i, False)
        Range("MinMSE2").Offset(i, 1).Value = Application.WorksheetFunction.VLookup(Range("MinMSE2").Offset(i, 0).Value, _
            Range(Range("ChartStocks").Offset(1, i - 1), Range("ChartStocks").Offset(1, 7).End(xlDown)), 9 - i, False)
    Next i

    Range(Range("ChartMonths").Offset(1, 0), Range("ChartMonths").Offset(1, NumStocks).End(xlDown)).Name = "Chart2Data"
    Worksheets("Option2").Activate
    ActiveSheet.ChartObjects("MSEChart").Select
    ActiveChart.SetSourceData Source:=Range("Chart2Data")
    For i = 1 To NumStocks
        ActiveChart.SeriesCollection(i).Name = Range("StockName2").Offset(i, 0).Value
    Next i
    With Worksheets("Option2")
        CalcDone = False
        .cmbStocks.Value = .cmbStocks.List(0)
        .optMSE = True
        CalcDone = True
        Call .optMSE_Click
    End With
    Range("A1").Select
    Application.ScreenUpdating = True
End Sub
```

Figure CS21.15 The second part of the *DoOption2* procedure.

The event procedures for the “Option2” sheet are for the output sheet’s controls (see Figure CS21.16 and CS11.17). Using the data on the calculation sheet, they update both the MSE chart and the histograms, based on the selected stock.

```
Sub cmbStocks_Change()
    If CalcDone Then
        optStock = True
        optMSE = False
        Application.ScreenUpdating = False
        Dim StockIndex As Integer
        StockIndex = cmbStocks.ListIndex + 1
        NumMon = Range("MinMSE2").Offset(StockIndex, -1).Value

        Worksheets("Calc2").Visible = True
        With Worksheets("Calc2")
            .Range(.Range("BetaEst").Offset(NumMon + 1, 0), .Range("BetaEst").Offset(146, 0)).FormulaR1C1 = _
                      "=SLOPE(R[-1]C[" & StockIndex - 8 & "]:R[-" & NumMon & "]C[" & StockIndex - 8 & "], R[-1]C[-9]:R[-" _
                      & NumMon & "]C[-9])"
            .Range(.Range("BetaEst").Offset(NumMon + 1, 0), .Range("BetaEst").Offset(146, 0)).Name = "HistInput"

            .Range(.Range("HistOutput"), .Range("HistOutput").Offset(20, 2)).Clear
            Application.Run "ATPVBAEN.XLA!Histogram", .Range("HistInput"), .Range(.Range("HistOutput"), _
                .Range("HistOutput").Offset(20, 2)), , False, True, , False
            .Range(.Range("HistOutput"), .Range("HistOutput").Offset(20, 0)).NumberFormat = "$,##.000"

            Worksheets("Option2").Activate
            ActiveSheet.Shapes("Hist").ZOrder msoBringToFront
            ActiveSheet.ChartObjects("Hist").Select
            ActiveChart.SetSourceData Source:=.Range(.Range("HistOutput"), .Range("HistOutput").Offset(0, 2).End(xlDown))
        End With
        Worksheets("Calc2").Visible = False
        Range("A1").Select
        Application.ScreenUpdating = True
    End If
End Sub
```

Figure CS21.16 The first part of the Option 2 output sheet code.

```
Sub optMSE_Click()
    If CalcDone Then
        If optMSE Then
            Application.ScreenUpdating = False
            ActiveSheet.Shapes("MSEChart").ZOrder msoBringToFront
            ActiveSheet.ChartObjects("MSEChart").Select
            ActiveChart.SetSourceData Source:=Worksheets("Calc2").Range("Chart2Data")
            For i = 1 To NumStocks
                ActiveChart.SeriesCollection(i).Name = Range("StockName2").Offset(i, 0).Value
            Next i
            Range("A1").Select
            Application.ScreenUpdating = True
            optStock = False
        End If
    End If
End Sub

Sub optStock_Click()
    If optStock Then
        optMSE = False
        Call cmbStocks_Change
    End If
End Sub
```

Figure CS21.17 The second part of the Option 2 output sheet code.

The navigational codes are for the “End” and “Return to Menu” buttons (see Figure CS21.18).

```
Sub EndProg()
    Worksheets("Welcome").Visible = True
    ActiveSheet.Visible = False
End Sub

Sub ReturnMenu()
    Worksheets("Welcome").Visible = True
    ActiveSheet.Visible = False
    frmOptions.Show
End Sub
```

Figure CS21.18 The navigational procedures.

Main	Initializes the application and retrieves the initial input from the users.
ClearPrev	Initializes the variables, clears previous values, and clears animation layout formatting.
PrepCalc	Prepares calculation and output sheets based on selected stocks.
Portfolio Form Code	Updates the array to note which stocks are in the users' portfolio.
Options Form Code	Directs the users to the appropriate output sheet.
DoOption1	Performs calculations for Option 1 and updates the output sheet.
Option 1 Output Sheet Code	Updates the chart based on changes to the output sheet controls.
DoOption2	Performs calculations for Option 2 and updates the output sheet.
Option 2 Output Sheet Code	Updates charts based on changes to the output sheet controls.
Navigational	Used for navigational buttons

CS21.5 *Re-solve Options*

The re-solve options for both Option 1 and Option 2 are provided by the controls on the worksheet. These controls allow the users to modify their input values and re-run the calculations by clicking on the "Calculate" button again after modifying the input values. On the Option 1 output sheet, the users can change the interval of months in order to view various optimal beta results and corresponding minimum MSE values. On the Option 2 output sheet, the users can alter the bounds for the trial number of months and view various values for the optimal number of months and corresponding MSE and mean beta values.

Option 1	The users can change the input values on the output sheet. To re-run calculations, they click on the "Calculate" button again.
Option 2	The users can change the input values on the output sheet. To re-run calculations, they click on the "Calculate" button again.

CS21.6 *Summary*

- The Beta of Stocks application seeks to predict the return on several stocks based on the return of the market. The users have the following two options for estimating the beta of stocks: find the optimal Beta value per stock for a given period of time; or evaluate the MSE for various values over the past months to use in Beta estimate.
- This application consists of five worksheets: the welcome sheet, two calculation sheets, and two option sheets.
- The application includes two user forms, controls on both of the output sheets, and several navigational and functional buttons.
- Several procedures for this application allow the users to …
- The re-solve options for both Option 1 and Option 2 are provided by the controls on the worksheet. These controls allow the users to modify their input values and re-run the calculations by clicking the “Calculate” button again after modifying the input values.

CS21.7 *Extensions*

- Allow the users to view the calculation sheets by creating appropriate navigational buttons and procedures.
- Add an additional re-solve option that allows the users to modify their portfolio at any time.
- Add an option to make stock return predictions based on new market return data and the found beta values.
- Add an option to compare the minimum MSE found from Option 1 and Option 2.